


Introduction to automated image processing using macros in Fiji



```
iterateOverFolder_v2.jm
File Edit Language Templates Run Tools Tabs
iterateOverFolder_v2.jm
1 // Written by d.haenni@zmb.uzh.ch and joana.delgadomartins@zmb.uzh.ch
2 // February 2019, Bio 407
3 // This is a template for processing entire folders
4 // putting the output to a folder of choice.
5
6
7 // Creates a string variable will be later used to filter out files that do not match this file extension.
8 suf="oib";
9
10 // Creates a string var. in which the path of the chosen input folder is stored.
11 inputDir=getDirectory("Select folder containing files to process");
12
13 // Creates a string var. in which the path of the chosen output folder is stored.
14 outputDir=getDirectory("Select an output folder");
15
16 // Batch mode allows processing of a macro without displaying images on the screen.
17 // This accelerates processing by a lot and keeps your screen clean.
18 setBatchMode(false); // set to true to activate or set to false to stay in displaying mode
19
20 // This closes all images that are open.
21 run("Close All");
22
23 // An array variable is created containing all the file names of the files within the input folder.
24 // An array is list of variables.
```

Run Batch Kill Show Errors Clear

Center for Microscopy and Image Analysis
University of Zurich

This script was written as a short summary for the users of Center for Microscopy and Image Analysis, University of Zurich, Switzerland.

April 2019, Dominik Hänni and Joana Delgado Martins.

This script was written using Fiji, ImageJ 1.52i.

Contents

1. ABBREVIATIONS AND CONCEPTS	6
2. REMARKS	6
3. SOFTWARE: FIJI	6
4. PLUGINS AND UPDATING	6
5. MACROS	7
6. SCRIPTING AND SUPPORTED LANGUAGES	9
7. COMMENTING	10
8. RUN SELECTED CODE	11
9. SAVING YOUR IMAGE	11
10. ADDING A SCALE BAR AND SAVING A SECOND IMAGE	12
11. VARIABLES, ADDING FLEXIBILITY TO YOUR MACRO	12
12. BUILT IN FUNCTIONS	13
13. STRING MANIPULATION AND FILENAMES	14
14. ITERATE OVER A FOLDER	14
FOR LOOPS	15
15. BATCH MODE	16
16. RUNNING THE TEMPLATE MACRO	16
17. LITERATURE AND FURTHER INFORMATION	19

1. Abbreviations and Concepts

ImageJ Macro: a script that automates a series of ImageJ commands and also allows for variables, control flow, etc.

2. Remarks

This script was written for practical training purposes by the Center for Microscopy and Image Analysis, University of Zurich. Theory is kept to a minimum.

The script should be used as a guideline for hands-on training.

Highlighting styles used in this script:

File names are set in green (e.g. [Cell_Division_Pinhole_0_5AU_Nyquist_7386.tif](#))

Step 1.

Practical steps are set in italic.

Keyboard shortcuts here

Macro code: `run("Enhance Contrast", "saturated=0.35");`

Or screenshots of the ImageJ macro editor.

3. Software: Fiji

This script gives a basic introduction to scientific image processing. All exercises can be done using ImageJ or **Fiji** which is just a version of ImageJ supplemented with many useful plugins. It can be downloaded for free from

<https://fiji.sc/>



Downloadable distributions are available for Windows, Mac OS X and Linux. It can read many image and microscope formats including TIFF, GIF, JPEG, BMP, DICOM, FITS, IMS, LIF, CZI, 'raw'.

Install your own copy of Fiji and selected plugins in your home folder if you are using the virtual machines available to the users of the Center of Microscopy and Image Analysis. Otherwise save it locally in your computer.

4. Plugins and Updating

When installing Fiji from the Downloads page, the latest *stable* version is available. However Fiji and its various plugins are frequently updated so you should check if you have the latest updates through

Step 1.

Help>Update

Here you can also manage the update sites and install additional useful plugins which are not installed by default.


5. Macros

Fiji macros are very useful tools not only to **automatize repetitive tasks**, but also to **standardize** and **document** your image processing efforts.

You can easily start recording the commands used in your image processing through the macro recorder

Step 2.

Plugins>Macro>Record...



Step 3.

Open an image from the provided folder for this training session.

Image1_Widefield_Actin.tif.

Create a copy and rename your image through

Step 4.

Image>Duplicate


Ctrl+Shift+ D

Choose a LUT of your choice for the selected image through

Step 5.

Image>Lookup tables

You will notice that each operation is being recorded in the macro recorder.




Step 6.

Adjust the brightness and contrast of you channels accordingly by using the “Set” button

Select Image>Adjust>Brightness/Contrast

Ctrl+ Shift+C



```
//run("Brightness/Contrast...");
```

```
setMinAndMax(230, 10000);
```


Step 7.


You can also automatically set these values by choosing “Auto”. In this case the following command will appear

```
run("Enhance Contrast", "saturated=0.35");
```

Step 8.

You can now create your own macro by pressing “Create” on the recorder window.





6. Scripting and Supported Languages

A scripting language is a programming language used to control another software such as a runtime environment or in this case ImageJ. Usually scripting languages are interpreted and not compiled. We will be using **ImageJ Macro language (IJM)** which is designed to be easy to read, learn and use. Programs written in the IJM, or macros, can be used to perform a sequence of actions in ImageJ.

ImageJ also supports several other scripting languages. To find more you can check here <https://imagej.net/Scripting>.

Like other programming languages, the IJM has basic structures that can be used for expressing algorithms. Those include variables, control structures (e.g. looping statements) and user-defined functions. In addition, the IJM provides access to the ImageJ functions available from the graphical user interface menu commands and to a large number of built-in functions aimed at working with the different data structures used in ImageJ (images and image windows, regions of interest, the results table, plots, image overlays, etc.).

Remark: Not all but most ImageJ plugins are scriptable. Only if the developer of a specific plugin included the scripting functionality properly it can be seen in the macro editor and used in your code.

In your macro editor, save your macro

Step 9.
File>Save As

You will realize that your text will appear formatted in a specific way (Syntax highlighting):

```

1 run("Green");
2 run("Cyan");
3 run("glow v2");
4 run("Yellow");
5 //run("Channels Tool...");
6 Stack.setDisplayMode("composite");
7 //run("Brightness/Contrast...");
8 run("Enhance Contrast", "saturated=0.35");
9 setMinAndMax(0, 40);
10 setMinAndMax(3, 182);
11 setMinAndMax(3, 38);
12
13

```

Commands will automatically appear in brown,

“text in magenta between inverted commas”,

values in blue and

comments in green.

All commands end with ;

You can check other available scripting languages under the

menu Language. Select IJ1 Macro Language in case your macro is shown in plain text.

At this point you probably also recorded unnecessary operations and mistakes.

Step 10.

You should now remove unnecessary operations from your code.

Step 11.

Now try to open the image

Image2_Widefield_Actin.tif

and apply the same settings using your recorded macro by pressing “Run”.

7. Commenting

It is essential not only for you but also for your successors that you correctly document your macro. Also include a header where you write

- who programmed the macro (also include a contact email)
- when
- where
- for what purpose
- assumptions, known limitations, data requirements, data preprocessing, needed plugins
- other useful information for running the macro

This will be helpful not only when you need to share the code with other colleagues, but also for your own recollection.


To add text that is only intended to be used for commenting your macro you can either include

// At the beginning of the line

Or

```
/*
*
*/
```

For multiple lines (once you've written `/*` and press Enter in your macro editor it will automatically fill the following 2 lines. You only need to write in front of the `*` and press Enter for a new line.




8. Run Selected Code

When debugging/optimizing your macro you might find it useful to run only selected parts of your code rather than the whole macro. You can do this in your macro editor through

Step 12.

Run>Run selected code


Ctrl+Shift+R



9. Saving your Image

Step 13.

Save your image as a tif file.



```
Recorder
Record: Macro Name: Macro.ijm Create ?
run("Red Hot");
//run("Brightness/Contrast...");
run("Enhance Contrast", "saturated=0.35");
saveAs("Tiff", "Z:/zmbcourse/ZMB_Bio_407/Image_Processing/Joana/Image1_Widefield_Actin_processed.tif");
```

You will realize that in the recorder the **exact path** will be given.

10. Adding a Scale Bar and Saving a Second Image

Record the commands for the following steps and add them to your macro:

Step 14.

Convert into RGB

Step 15.

*Add a scale bar through
Analyze>Tools>Scale Bar...*

Step 16.

Save the second image with a different name.

11. Variables, adding Flexibility to your Macro

You would now love to apply this macro to a whole folder and save the corresponding images in an output folder. But how?

The command recorder writes down exactly what you do, including the names of the images you use as well as the absolute paths. To use a more general macro on more than one image, specific names must be replaced with either generic names or **variables**. You can store your filenames as variables that can be later called in the macro and used for other operations.

The ImageJ macro language is typeless. Variables do not need to be declared and do not have explicit data types. They are automatically initialized when used in an assignment statement. A variable can contain a number, a boolean, a string or an array. Numbers are stored in 64-bit double-precision floating point format. Booleans are represented by the numbers 0 and 1. The same variable can be any of these things at different times.

Strings are important variables to work with in the ImageJ macro language. They are used to handle file names and file paths, window titles and user interaction messages, but also to feed arguments to built-in commands.


Variable names are case-sensitive. "Name" and "name" are different variables.

Step 17.

Store and print a variable "testvariable" with your text of choice.

```
//Variables|
testvariable="Hello. this is a test variable";
print(testvariable);
```

This will be shown in the Log window.



Step 18.

Store a variable "testvariable2" with a different text. Print it.

Be careful when assigning variables as you might overwrite previously defined ones. Choose informative variable names and a coherent naming scheme that you can easily identify or debug later on.

12. Built in Functions

Besides the commands recorded in the macro recorder ImageJ also has many useful built in functions such as

```
getTitle();
```

This function allows you to access the name of your image.

Step 19.

Store a variable "image" which contains your image title by adding


```
image=getTitle();
```

Step 20.

As the output of such instructions is not always visible you can use the command print so you can easily keep track. This will be shown in the Log window.

```
print("The name of the image is "+image);
selectWindow(image);
```

You can also print variables and additional text by using the + sign as you can see in the example. Remember your text must be between ". You will easily recognize this due to its different syntax highlight (magenta).



You will also find many useful built in functions in

<https://imagej.nih.gov/ij/developer/macro/functions.html>

13. String Manipulation and Filenames

Additionally you can change your string such as "image" by using different in-built functions:

`indexOf(string, substring)`

Returns the index within string of the first occurrence of substring.

`substring(string, index1, index2)`

Returns a new string that is a substring of string. The substring begins at index1 and extends to the character at index2 - 1.

`replace(string, old, new)`

Returns the new string that results from replacing all occurrences of old in string with new, where old and new are single character strings.

`toString(number)`

Returns a decimal string representation of number.

Very often it is necessary to deal with filenames using the following functions:

`File.getName(path)`

Returns the last name in path's name sequence.

`File.getParent(path)`

Returns the parent of the file specified by path.

Step 21.

Use ***replace(string, old, new)*** to change string ***image*** containing your file name

```
newname=replace(image,"Actin","Actin2");  
print("new image name is " + newname);
```

14. Iterate over a Folder


Step 22.

Close the macro recorder.

Drag & Drop the file `IterateOverFolder_template_empty_ZMB_201904.ijm`

Step 23.

Take a look at the comments and structure of the template.



Using additional built in functions will help you get the path for your input and output folders.

```
// Creates a string var. in which the path of the chosen input folder is stored.
inputDir=getDirectory("Select folder containing files to process");

// Creates a string var. in which the path of the chosen output folder is stored.
outputDir=getDirectory("Select an output folder");
```

The variable "list" will be used to store all the file names inside the input directory:

```
list = getFileList(inputDir);
```

For loops

In order to go through all the files contained in your folder we will need to create an **iteration loop**. Looping statements are used to repeatedly execute a block of code.

Remark: Be careful that you only have the images you want to process in your folder. Otherwise you need to include additional code which instructs ImageJ regarding the types of files that will be used.

The ImageJ macro language has several loop statements for controlling the flow of the processing. In this case we will set up a **for** loop.

The *for* statement has the form:

```
for (initialization; condition; increment) {
    statement(s)
}
```

You will find this structure in the template:

```

list = getFileList(inputDir);

for (i=0; i<list.length; i++) {

    run("Bio-Formats Importer", "open=[" + inputDir + list[i] + "]" color_mode=De
title=getTitle());

    // Enter your code here //////////

    // END of inserted code  //////////

    saveTitle= outputDir+File.separator+"processed_"+title;

    saveAs("Tiff", saveTitle);

}

```

By adding `.length` to your “list”-variable the `for` loop knows how many times to iterate (the length of your file list). Alternatively you could manually select how many files you wish to process by placing a number instead of `list.length`.

15. Batch Mode

If you do not need to visualize each image as it is being processed you can

Step 24.

Set Batch mode to `true`.

```
setBatchMode(true);
```

The batch mode allows processing of a macro without displaying the images on the screen. This improves the performance of your macro dramatically and keeps your screen clean.

If you are running your macro in batch mode, print commands to the console will be particularly helpful to follow what is going on and to debug your code.

As good practice always set Batch Mode to false again

```
setBatchMode(false);
```

at the end of your macro.

16. Running the Template Macro

You can now try to apply your instructions (do not include the save options) to a test input folder.


Step 25.

Try applying the same B&C settings to all the images inside the folder by copying your code into the section


```
// Enter your code here //////////
```


Step 26.

Run the macro.




You will be asked to choose an input folder and an output folder.



After you choose your origin and destination folders the macro will store these paths and try to open the files inside.

In the Log window you will be able to see how many files your input folder contains.



Step 27.

*How would you proceed if you would like to save two images?
One with and one without the scale bar?*

Open the macro [IterateOverFolder_B&C_scale_2images_ZMB_201904.ijm](#)

In this example, after the B&C settings are adjusted the image is duplicated.

The name of the second image is stored under "title2".

To save each image the command `selectWindow()` allows selection of the appropriate window.

```
run("Duplicate...", " ");
title2=getTitle();
```

```
//Here you add the scale bar
```

```
run("Scale Bar...", "width=50 height=16 font=56 color=White background=Non
```

```
// END of inserted code ////////////////
```

```
//Saving first image
```

```
selectWindow(title);
saveTitle= outputDir+File.separator+"noscalebar_"+title;
print("new file path is "+saveTitle);
```

```
saveAs("Tiff", saveTitle);
```

```
close();
```

```
//Saving second image
```

```
selectWindow(title2);
saveTitle= outputDir+File.separator+"withscalebar_"+title;
print("new file path is "+saveTitle);
```

```
saveAs("Tiff", saveTitle);
```

```
close();
```

```
}
```

```
setBatchMode(false);
```

```
print ("-----All is done!-----");
```

17. Literature and Further Information

- Image J Macro language
 - <https://imagej.nih.gov/ij/developer/macro/macros.html>
- Introduction to Macro Programming
 - https://imagej.net/Introduction_into_Macro_Programming
- The ImageJ Built-in Macro Functions
 - <https://imagej.nih.gov/ij/developer/macro/functions.html>
- You will also find various forums and channels where you can place your questions and find solutions.
 - <https://imagej.net/Help>

Ways to get help

There are several popular communication channels:

Forum	The recommended way to get help. Very active. Rich modern interface. Public, archived discussion.
Mailing Lists	An established and trusted resource for help. Thousands of subscribers. Public, archived discussion.
Chat	ImageJ developers and experts frequent real-time chat services including Gitter and IRC.
Stack Overflow	Some people like to post about ImageJ on Stack Overflow, a programming Q&A site.
Reddit	Some people like to post about ImageJ on Reddit, a community bulletin board.